
Simulation mit Spielkarten und mit dem TI83 Ac

Bezugsaufgabe:   Schroedel-Buch Seite 457, Aufgabe 9.

3-mal 6 Spielkarten (jeweils nummeriert von 1 bis 6) sind vorhanden.
6 Spieler wählen zunächst für sich eine „Glückszahl“ . 
Ein Spielleiter zieht die Karten ohne Zurücklegen ! 
Gewonnen hat derjenige, der als erster alle 3 Glückszahlen zusammen hat.

Frage:  Wie häufig kommt es vor, dass ein Spieler gewonnen hat und gleichzeitig bei 
mindestens  einem der anderen Spieler noch kein einziges Exemplar seiner Glückszahl gezogen 
wurde ?

Simulieren kann man das Spiel, indem man für die 6 Karten  7;8;9;10;Bube;Dame jeweils 3 
verschiedene Farben auswählt, z.B.  Kreuz, Pik, Herz . Die Karten werden gemischt und nacheinander 
vom Stapel gezogen . Jeder Schüler kann nun mit einem reduzierten Kartenspiel diese Simulation 
mehrmals durchführen. 

2 mögliche Spielergebnisse sind hier zu sehen .

 

Im ersten Spiel hat jeder Spieler mindestens 1-mal seine Glückszahl, jedoch im zweiten Spiel fehlen 
sowohl 7 als auch 8 und Dame . Dort ist also der zu untersuchende Fall eingetreten.

Man kann nun z.B. 100-mal spielen und zählen, wie oft der zu untersuchende Fall eintritt. Für jeden 
eingetretenen Fall schreibt man z.B. eine 1 in eine Tabelle, für jeden nicht eingetretenen Fall eine 0. 
Die Summe der 1en ergibt dann die absolute Häufigkeit H.  
Die relative Häufigkeit h = H/100 ist dann ein Schätzwert für die Wahrscheinlichkeit  
P(„bei mindestens einem Spieler ist am Spielende seine Glückszahl noch nicht gezogen worden“)

Wie groß ist P ??

Simulation mit dem TI 83

Man erzeugt mit dem Befehl   randInt(1,18,25) eine Serie von 25 Zufallszahlen im Bereich [1;18] .
Für 25 kann auch eine größere Zahl eingesetzt werden, falls nötig.
Man benötigt dann eine Zuordnung der 18 möglichen Zahlen (Karten) zu den 6 Zahlen 1;2; … ;6 .
Eine mögliche Zuordnung ist
1;7;13    →  1
2;8;14    →  2
3;9;15    →  3
4;10;16  →  4
5;11;17  →  5
6;12;18  →  6

Anmerkung:  Automatisch lässt sich diese Zuordnung mittels   Z – 6*int((Z-1)/6)   erzeugen .
Dabei bedeutet int(X) die größte ganze Zahl , die kleiner oder gleich X ist .
Z.B.:        int(5,3)=5     int(12,99)=12     int(7)=7     int(-2,34)=-3 usw.

In der Serie darf jede der Zahlen 1 bis 18 nur höchstens 1-mal gezählt werden !
Es handelt sich also um ein Ziehen ohne Zurücklegen !
Das Spiel bricht ab, wenn eine der Zahlen 1 bis 6 genau 3-mal vorgekommen ist . Man muss dann 
untersuchen, ob mindestens eine der anderen Zahlen überhaupt nicht vorgekommen ist.

Beispiele ( eingeklammert werden die mehrfach vorkommenden Zahlen im Bereich [1;18] . 

11   8   10   4   18   9   2   13   14 „2“ gewinnt
 5    2     4   4    6    3   2     1     2     Alle Zahlen sind vorgekommen

 8   15   11   14   (15)   5   6   2 „2“ gewinnt
 2     3     5    2     (3)    5   6   2        „1“ und „4“ sind nicht vorgekommen

10   (10)   8   1   2   13   12   (12)   3   6   (3)   (2)   5   15   14 „2“ gewinnt
 4     (4)    2   1   2     1     6    (6)    3   6   (3)   (2)   5     3     2   Alle Zahlen vorgekommen

 6   8   13   17   10   15   14   9   (14)   5   (8)   1   (5)   (10)   16   (14)   2 „2“ gewinnt
 6   2     1     5    4     3     2    3    (2)    5   (2)   1   (5)    (4)     4     (2)    2 Alle Zahlen vorgekommen

 8   10   1   13   16   17   (16)   3   (1)   6   (17)   (16)   (16)   (13)   15   7 Die „1“ gewinnt
 2    4    1     1    4     5      (4)   3   (1)   6    (5)     (4)     (4)     (1)     3    1   Alle Zahlen vorgekommen

14   16   11   2   (2)   (11)   5   13   18   (2)   17 Die „5“ gewinnt
  2    4     5    2   (2)    (5)    5    1     6    (2)     5    „3“ ist nicht vorgekommen 

Bis zu diesem Zeitpunkt sind 2 von 6 Spielen  „positiv“ (im Sinne der Fragestellung) ausgegangen. Also 
liegt die rel. Häufigkeit hier bei etwa 33% . Dieser Schätzwert ist noch sehr schlecht !
 



 Programmierung der Simulation mit dem TI83

Folgende Variablen werden verwendet:

L1:  Es werden die Speicher L1(1) bis L1(18) verwendet. Steht in L1(k) eine 1, so bedeutet es, dass
            die Zahl k vorgekommen ist. 0 bedeutet „nicht vorgekommen“.
L2:  Es werden die Speicher L2(1) bis L2(6) verwendet. In L2(k) steht die Anzahl des Auftretens der 
            Zahl k . Ist diese Zahl = 3, so ist das Spiel zu Ende.
N Anzahl der Versuchsdurchläufe
K Äußere Schleife von 1 bis 92   (Anzahl der durchgeführten Spiele).
I Innere Schleifen von 1 bis 6 oder von 1 bis 18 .
H Absolute Häufigkeit für das Eintreten des zu untersuchenden Ereignisses .
Z Zufallszahl, die gerade erzeugt wurde .
X Hilfsvariable; ordnet die Zahlen 1; 2; … ; 18  den Zahlen 1; 2; 3; 4; 5; 6 zu .
A Boolesche Variable für Abbruch ( 1=Abbruch; 0=kein Abbruch) .
Y Boolesche Variable; 1 bedeutet: „mindestens eine Zahl von 1 bis 6 ist nicht aufgetreten“ .

PRGM  „Simul  1“  

ClrAllLists
Prompt N
0 → H
For(K,1,N)
   For(I,1,18)
      0 → L1(I)
   End
   For(I,1,6)
      0 → L2(I)
   End
   0 → Y
   0 → A
   Repeat A=1
       randInt(1,18) → Z
       Z-6*int((Z-1)/6) → X
       If L1(Z)=0
       Then
          1 → L1(Z)
          1+L2(X) → L2(X)
       End   { von then }
       L2(X)=3 → A
   End   { von Repeat }
   For(I,1,6)
      If  L2(I)=0
      1 → Y
   End
   If  Y=1
       1+H → H
End   { von For K … }
Disp H/N

In L1 steht, welche Zahl (von 1 bis 18) beim letzen 
Simulationsdurchlauf vorgekommen ist und welche nicht.
L1(2)=0 bedeutet:   Die „2“ ist nicht vorgekommen.
L1(4)=1 bedeutet:   Die „4“ ist vorgekommen.

In L2 stehen die Anzahlen der beim letzen Simulationsdurchlauf 
vorgekommenen Zahlen (von 1 bis 6).
L2(1)=0 heißt: Die „1“ kam 0-mal vor.
L2(4)=3 heißt: Die „4“ kam 3-mal vor (Sieger).

Man erkennt eine rel.Häufigkeit von   h = 56,5% !
Dies kann als Schätzwert für die gesuchte Wahrscheinlichkeit P 
aufgefasst werden.

Achtung:    Sehr lange Rechenzeit !

Anmerkung zum Programm:
Statt int kann auch iPart (ganzz. Anteil) verwendet werden !



Warum lässt sich das Problem nicht mit   randInt(1,6) lösen ??

Antwort:     Bei randInt(1,6) ist die Einzelwahrscheinlichkeit jeder der Zahlen 1:2:…;6 immer 3/18=1/6, 
auch noch nach der 15. Ziehung, während sich bei dem zu untersuchendem Problem nach jeder Ziehung 
die Einzelwahrsch. ändert !  Ziehe ich etwa mit dem ersten Zug eine „5“, so beträgt beim nächsten Zug 
die W. für „5“ nicht mehr 3/18=1/6, sondern nur noch 2/17 . 

Die „falsche“ Simulation durchgespielt mit jeweils  randInt(1,6,30) :

( Es wird hier vorausgesetzt, dass die Kugeln mit Zurücklegen gezogen werden)

Ergebnisse:

4   3   1   6   3   4   5   2   2   1   4 „4“ gewinnt
Alle Zahlen vorgekommen

4   2   2   2 „2“ gewinnt
1;3;5;6 nicht vorgekommen

2   1   2   1   2   „2“ gewinnt
3;4;5;6 nicht vorgekommen

4   1   1   6   6   1
6   4   3   1   3   3
2   5   6   2   5   2

Man erkennt, dass der zu prüfende Fall recht häufig auftritt ( hier   h = 5/6  bei N=6 )

Wie groß ist P ?? Lösung:   P ≈ 0,858           ( mit dem folgendem Programm prüfen !)

PRGM  „Simul2“

ClrAllLists
Prompt N
0 → H
For(K,1,N)
   For(I,1,6)
      0 → L2(I)
   End
   0 → Y
   0 → A
   Repeat A=1
       randInt(1,6) → X
      1+L2(X) → L2(X)
       L2(X)=3 → A
   End   { von Repeat }
   For(I,1,6)
      If  L2(I)=0
      1 → Y
   End
   If  Y=1
       1+H → H
End   { von For K … }
Disp H/N


